



# Grundlagen der Informatik 1

## WS 2008/09

Prof. Mühlhäuser, Dr. Rößling, Melanie Hartmann, Daniel Schreiber  
<http://proffs.tk.informatik.tu-darmstadt.de/gdi1>

### Übung 2 Lösungsvorschlag v1.0

27.10.2008

---

## 1 Mini Quiz

- ☒ Elemente einer Liste können von verschiedenen Datentypen sein.
- ☒ Rekursive Datentypen und Prozeduren brauchen zum Terminieren immer einen Rekursionsanker.
- ☐ Man kann mit `cons` Listen erzeugen, die man mit `list` nicht erzeugen kann.
- ☒ Strukturdefinitionen erzeugen automatisch Konstruktoren, Selektoren und Prädikate.
- ☐ Natürliche Zahlen sind abgeschlossen unter der Subtraktion.

## 2 Fragen

1. Beschreiben Sie mit eigenen Worten, was ein rekursiver Datentyp ist.
2. Erklären Sie wieso rekursive Datenstrukturen in der Praxis nicht unendlich groß werden.
3. Welche zwei Ansätze beim Design von Programmen mit Hilfe von Wunschlisten sind Ihnen bekannt? Diskutieren Sie Vor- und Nachteile des jeweiligen Ansatzes.
4. Erklären Sie mit eigenen Worten, was Abgeschlossenheit ist.

### Lösungsvorschlag:

1. *Eine Rekursion ist der Verweis auf sich selbst. Ein Datentyp ist rekursiv, wenn bei seiner Definition der Datentyp selbst verwendet wird. Zum Beispiel ist ein Strukturtyp, bei dem der Typ eines Elementes wieder der Strukturtyp ist, rekursiv.*
2. *Rekursive Datenstrukturen verwenden stets einen Rekursionsanker, d.h. einen möglichen Wert für ein Element, das selber nicht wieder die Struktur enthält. Bei der Definition der Liste ist dies z.B. die leere Liste, die selber nicht wieder eine Liste enthält. Dies wird durch die Verwendung von heterogenen Daten modelliert. Würde die Definition nicht die leere Liste als Wert für die Struktur enthalten, so ließe sich niemals eine Liste erstellen, da immer weiter Elemente hinzugefügt werden müssten.*

3. *Ausgehend von der Wunschliste aller zu implementierenden (Hilfs-)Prozeduren existieren zwei komplementäre Ansätze fuer die Implementierung des Programmes. Das erste Verfahren ist der sog. Bottom-Up-Ansatz: Es werden zuerst diejenigen Prozeduren implementiert, die von keiner anderen Prozedur auf der Wunschliste abhängen. Der Vorteil besteht darin, dass jede implementierte Prozedur sofort ausgeführt und getestet werden kann. Der Nachteil ist, dass meist nicht klar ist, welche Prozeduren auf der untersten Ebene benötigt werden. Außerdem führt dieser Ansatz oft dazu, dass man sich in Details verliert und nicht mehr das Gesamtziel vor Augen hat.*

*Beim sog. Top-Down-Ansatz wird zuerst die Hauptprozedur der Wunschliste implementiert, und dann die dort aufgerufenen Prozeduren. Der Vorteil bei dieser Methode besteht darin, dass man sich schon zu Beginn auf das Hauptproblem konzentrieren kann, ohne an Details zu denken. Die Details werden dann nach und nach (inkrementell) in Form von Hilfsprozeduren ergänzt. Der Nachteil des Top-Down-Ansatzes ist, dass erst spät getestet werden kann. Man behilft sich oft dadurch, indem man leere Hilfsprozeduren anlegt, die zwar aufgerufen werden können, jedoch nur einen festen Wert zurückliefern. Oft ist jedoch eine Mischung aus beiden Ansätzen sinnvoll (Siehe Folien T3.28ff).*

4. *Wir sagen: Eine Menge von Elementen ist abgeschlossen bezüglich einer Operation, wenn die Anwendung der Operation auf Elementen der Menge wieder ein Element der Menge produziert. Beispiel: Die Menge der natürlichen Zahlen ist abgeschlossen bezüglich Addition und Multiplikation, da eine Summe oder ein Produkt zweier natürlicher Zahlen wieder eine natürliche Zahl ist. (Siehe Folien T3.10-3.13)*

### 3 Strukturen (K)

Versuchen Sie bitte die Aufgaben erst ohne einen Rechner zu lösen.

Gegeben sei folgende Strukturdefinition:

```
(define-struct mypair (first second))
```

Werten Sie folgende Ausdruck aus und geben Sie das Ergebnis an.

1. `(make-mypair 'a 'b)`
2. `(mypair? (make-mypair 'a 'b))`
3. `(mypair? (list 'a 'b))`
4. `(make-mypair 1 (make-mypair 2 empty))`
5. `(* (mypair-second (make-mypair 1 2)) (mypair-first (make-mypair 3 4)))`

**Lösungsvorschlag:**

1. *Konstruktoren sind selbstauswertend* `(make-mypair 'a 'b)`
2. `true`
3. `false`
4. *Konstruktoren sind selbstauswertend* `(make-mypair 1 (make-mypair 2 empty))`
5. `6`

## 4 Listen (K)

Nehmen Sie für die folgenden Aufgaben das Sprachlevel „Anfänger mit Listen-Abkürzungen“ an. Versuchen Sie bitte die Aufgaben erst ohne einen Rechner zu lösen. Diese Aufgabe muss bearbeitet werden, um die Hausübung lösen zu können.

1. Welche der folgenden Ausdruckspaare werden zu äquivalenten Listen ausgewertet?

- a) `(cons 1 (cons 2 (cons 3 empty)))` und `(list 1 2 3 empty)`
- b) `(cons (list '()) empty)` und `(list 'list empty)`
- c) `(list 7 '* 6 '= 42)` und `(cons 7 (cons '* (cons 6 (cons '= (list 42)))))`
- d) `(cons 'A (list '(I)))` und `(list 'A (cons 'I empty))`

### Lösungsvorschlag:

- a) *Die Listen sind nicht äquivalent. Die zweite Liste enthält als viertes Element die leere Liste, die erste Liste enthält nur drei Elemente. Die Auswertung ergibt:*  
`(list 1 2 3)` und `(list 1 2 3 empty)`
- b) *Die Listen sind nicht äquivalent. Das innere Vorkommen von list wird bei der ersten Liste ausgewertet, bei der zweiten Liste nur als Symbol verwendet. Die Auswertung ergibt:*  
`(list (list empty))` und `(list 'list empty)`
- c) *Die Listen sind äquivalent. Die Auswertung ergibt:*  
`(list 7 '* 6 '= 42)`
- d) *Die Listen sind äquivalent. Die Auswertung liefert:* `(list 'A (list 'I))`

2. Werden folgende Ausdrücke fehlerfrei ausgewertet? Falls nicht, begründen Sie bitte was zum Fehler führt.

- a) `(cons 1 (cons 2 (cons 3)))`
- b) `(cons 1 (list 2 (list '(3 + 4))))`
- c) `(list (cons empty 1) (cons 2 empty) (cons 3 empty))`


### Lösungsvorschlag:

- a) *Auswertung nicht möglich, da innerster cons Konstruktor nur einen statt zwei Parametern übergeben bekommt.*
- b) *Fehlerfreie Auswertung.*
- c) *Auswertung nicht möglich, da erster cons Konstruktor als zweiten Parameter eine Liste (und keine Zahl) erwartet.*

3. In der Vorlesung haben Sie erfahren, wie man Listen in Kästchenschreibweise darstellt (T3.8). Stellen Sie folgende Listen in Kästchenschreibweise dar:

- a) `(define A (list (cons 1 empty) (list 7) 9))`
- b) `(define B (cons 42 (list 'Hello 'world '!)))`

### Lösungsvorschlag:



4. Zu welchen Werten werden folgende Ausdrücke (A und B wie oben definiert) ausgewertet?  
Falls ein Auswertung nicht möglich ist, begründen Sie bitte.

- `(first (rest A))`
- `(rest (first A))`
- `(first empty)`
- `(append (first B) (rest (rest A)) (first A))`

#### Lösungsvorschlag:

- `(first (rest (list (cons 1 empty) (list 7) 9))) = (first (list (list 7) 9)) = (list 7)`
- `(rest (first (list (cons 1 empty) (list 7) 9))) = (rest (cons 1 empty)) = empty`
- Auswertung nicht möglich, da first eine nicht-leere Liste als Parameter erwartet.*
- Da (first B) die Zahl 42 zurückliefert und append nur Listen als Parameter verarbeiten kann, wird die Auswertung mit einem Fehler unterbrochen.*

5. Im folgenden sollen Sie rekursive Prozeduren auf Listen definieren. Vergessen Sie nicht, zuerst Vertrag, Beschreibung und ein Beispiel anzugeben.

- Schreiben Sie eine Prozedur `list-length: lst → num`, die eine Liste als Parameter erhält und die Länge der Liste zurückliefert. (Hinweis: Die leere Liste enthält keine Elemente. Verwenden Sie Rekursion, um das Ergebnis für nicht-leere Listen zu berechnen.)

#### Lösungsvorschlag:

```

1  ;; Contract: list-length: lst → num
2  ;; Purpose: computes length of a list
3  ;; Example: (list-length '(a b 0 1)) → 4
4  (define (list-length lst)
5    (if (empty? lst)
6        ;; list empty? anchor reached,
7        ;; do not recurse further
8        ;; empty list has zero elements so return 0
9        0
10       ;; else, list contains at least one element

```

```

11      ;; add one to the length of (rest lst)
12      ;; computed by the recursive call
13      ;; [(rest lst) is the list without the first element ]
14      (+ 1 (list-length (rest lst))))
15  ;; Test
16  (list-length '(1 2 3 4 5))
17  ;; should be
18  5

```

- b) Schreiben Sie eine Prozedur `member?: los symbol -> boolean`, die eine Liste von Symbolen und ein Symbol als Parameter erhält und zurückliefert, ob das Symbol in der Liste enthalten ist. (Hinweis: Die leere Liste enthält das Symbol sicher nicht. Verwenden Sie Rekursion, um das Ergebnis für nicht-leere Listen zu berechnen. Verwenden Sie die Prozedur `symbol=?: symbol symbol -> boolean`, um zu überprüfen ob zwei Symbole identisch sind.)

**Lösungsvorschlag:**

```

1  ;; Contract: member?: los symbol -> boolean
2  ;; Purpose:  computes if symbol is member of los
3  ;; Example: (member? '(a b) 'b) -> true
4  (define (member? los symbol)
5    (cond
6      ;; list empty? anchor reached,
7      ;; do not recurse further
8      ;; symbol not contained in empty list, return false
9      [(empty? los) false]
10     ;; else, list contains at least one element
11     ;; return true if the first element is equal
12     ;; to searched symbol
13     [(symbol=? (first los) symbol) true]
14     ;; first element is not the searched symbol,
15     ;; check if searched symbol is member of
16     ;; the rest of list by the recursive call
17     [else (member? (rest los) symbol)]))
18
19  ;; Test
20  (check-expect (member? '(a b c) 'c) true)

```

- c) Schreiben Sie eine Prozedur `remove-duplicates: los -> los`, die eine Liste von Symbolen als Parameter erhält und alle Duplikate aus der Liste entfernt.

Beispiel: `(remove-duplicates '(a a b) -> '(a b))`.

Verwenden Sie Rekursion und betrachten Sie drei Fälle:

- Die leere Liste.
- Das erste Element der Liste kommt im Rest der Liste vor.
- Das erste Element kommt nicht im Rest der Liste vor.

**Lösungsvorschlag:**

```

1  ;; Contract: remove-duplicates los -> los
2  ;; Purpose:  removes all duplicates from a list of symbols
3  ;; Example:  (remove-duplicates '(a a b)) -> '(a b)

```

```

4 (define (remove-duplicates list-of-symbols )
5   (cond
6     ;; list empty? anchor reached,
7     ;; do not recurse further
8     ;; empty list does not contain duplicates
9     ;; and can be returned
10    [(empty? list-of-symbols) list-of-symbols]
11    ;; first element is a duplicate
12    ;; (is a member of the rest of the list)
13    [(member? (rest list-of-symbols) (first list-of-symbols))
14     ;; so return rest of the list with remaining
15     ;; duplicates removed
16     (remove-duplicates (rest list-of-symbols))]
17    ;; first element is not a duplicate.
18    ;; Remove duplicates from the rest of the list
19    ;; and add first element at the beginning
20    [else (cons
21            (first list-of-symbols)
22            (remove-duplicates (rest list-of-symbols)))]])
23
24 ;; Test
25 (check-expect (remove-duplicates '(a a b)) '(a b))

```

## Hausübung

Die Vorlagen für die Bearbeitung werden im Gdl1-Portal bereitgestellt. Kommentieren Sie Ihren selbst erstellten Code. Die Hausübung muss bis zum Abgabedatum im Gdl1-Portal abgegeben werden. Der Fachbereich Informatik misst der Einhaltung der Grundregeln der wissenschaftlichen Ethik großen Wert bei. Zu diesen gehört auch die strikte Verfolgung von Plagiarismus. Mit der Abgabe Ihrer Hausübung bestätigen Sie, dass Sie bzw. Ihre Gruppe alleiniger Autor des gesamten Materials sind. Falls Ihnen die Verwendung von Fremdmaterial gestattet war, so müssen Sie dessen Quellen deutlich zitieren.

**Abgabe: Spätestens Fr, 07.11.08, 16:00.**

**Wichtig: Um die volle Punktzahl zu erlangen, müssen Sie jede Ihrer Prozeduren und Hilfsprozeduren mindestens mit Vertrag, Beschreibung und Beispiel kommentieren, sowie jeweils Testfälle angeben.** Verwenden Sie als Sprachlevel für die gesamte Hausübung „Anfänger mit Listen-Abkürzungen“.

## 5 EMail Adresse

1. Tragen Sie eine gültige EMail Adresse im Gdl Portal ein („Mein Konto“->Bearbeiten), sonst kann es sein, dass wichtige eMails Ihres Tutors Sie nicht erreichen!!
2. Lesen Sie entweder regelmäßig die EMail in ihrer RBG Mailbox oder richten Sie sich einen Mail forward ein. Sollten Sie Probleme damit haben, folgen Sie den Hinweisen hier:  
[www.rbg.informatik.tu-darmstadt.de/index.php?id=560#forward](http://www.rbg.informatik.tu-darmstadt.de/index.php?id=560#forward).

## 6 Listen und Strukturen für Suchmaschinen (13P)

In dieser Aufgabe werden Teile der Funktionsweise einer Internet-Suchmaschine auf sehr einfache Art und Weise nachgebaut.

Wir beginnen mit der Definition einer Ähnlichkeitsfunktion, der Cosinus-Ählichkeit, die wir später benötigen:

```
(define (cosine-similarity vector1 vector2)
  (/
    (vec-mult vector1 vector2)
    (*
      (sqrt (vec-mult vector1 vector1))
      (sqrt (vec-mult vector2 vector2))))))
```

Diese Funktion verwendet die Funktion `vec-mult`, die das Skalarprodukt zweier Vektoren berechnen soll. Vektoren werden als Listen von Zahlen dargestellt. Zur Erinnerung: Das Skalarprodukt zweier Vektoren ist die Summe der Produkte der Komponenten der Vektoren. Also,

$$(\text{vec-mult } '(1\ 1\ 0) \ '(1\ 0\ 1)) = (+\ (*\ 1\ 1)\ (*\ 1\ 0)\ (*\ 0\ 1)).$$

1. (K) Schreiben Sie die Prozedur `vec-mult`, die zu zwei Vektoren (Listen von Zahlen) das Skalarprodukt berechnet.

Hinweis: Multiplizieren Sie die jeweils ersten Komponenten der Vektoren und addieren Sie das Ergebnis zum Skalarprodukt der restlichen Vektorkomponenten. Das Skalarprodukt zweier leerer Vektoren ist 0. Sie dürfen voraussetzen, dass die Vektoren gleich viele Komponenten haben.

- Berechnen Sie `(cosine-similarity '(1 0) '(0 1))`
- Berechnen Sie `(cosine-similarity '(1 1) '(1 1))`

#### Lösungsvorschlag:

s.u. Zeile 1 - 35

2. (K) Definieren Sie eine Struktur zum Speichern von Dokumenten. Ein Dokument hat einen Namen und einen Inhalt. Der Inhalt soll hier eine Liste von Symbolen sein (die Wörter des Dokuments, z.B. `'(Dies ist der Inhalt eines Dokuments)`). Definieren Sie zwei Dokumente:
  - `doc1` mit dem Namen `'doc1` und dem Inhalt `'(mouse keyboard screen)`.
  - `doc2` mit dem Namen `'doc2` und dem Inhalt `'(pascal java scheme)`.

#### Lösungsvorschlag:

s. u. Zeile 37 - 50

3. Schreiben Sie die Funktion `index`, die einen Index zu einer Liste von Dokumenten erzeugt. Ein Index ist eine Liste aller im Inhalt der Dokumente vorkommenden Symbole ohne Duplikate. Verwenden Sie für Dokumente die Struktur aus Aufgabe 6.2. Hinweis: Eine leere Liste von Dokumenten ergibt einen leeren Index. Ist die Liste der Dokumente nicht leer, so kann der Inhalt des ersten Dokumentes mit der Prozedur `append` an den Index der restlichen Dokumente angefügt werden. Zur Erinnerung: Die Prozedur `append` verbindet mehrere Listen, z.B. `(append '(a b) '(c d)) = '(a b c d)`. Verwenden Sie die Prozedur `remove-duplicates` aus der Präsenzübung um Duplikate aus dem Index zu entfernen.
  - Berechnen Sie `(index (list doc1 doc2))`.

- Berechnen Sie `(index (list doc1 doc1 doc2))` und überprüfen Sie, dass keine Duplikate vorkommen.
- Definieren Sie `myindex` als `(index (list doc1 doc2))`.

**Lösungsvorschlag:**

s. u. Zeile 51 - 82

Der Index aus der vorigen Aufgabe wird verwendet, um Anfragen und Dokumente in Wortvektoren umzuwandeln. Ein Wortvektor zu einem Dokument oder einer Anfrage enthält für jedes Element des Index eine 0 oder eine 1. Eine 0 bedeutet: Das entsprechende Wort aus dem Index kommt nicht im Dokumentinhalt / der Anfrage vor, eine 1 bedeutet es kommt vor. Dabei spielt es keine Rolle, dass im Dokumentinhalt / der Anfrage auch Symbole vorkommen, die nicht im Index sind.

Beispiel: Der Index sei `'(this list is an index)` und die Anfrage `'(this is a query)`, so ist der zugehörige Wortvektor `'(1 0 1 0 0)`. Dies bedeutet, dass `'this` und `'is` in der Anfrage vorkommen, alle anderen Symbole aus dem Index nicht.

4. Schreiben Sie eine Funktion `word-vector`, die eine Liste von Symbolen (die Anfrage) oder ein Dokument übergeben bekommt sowie eine zweite Liste von Symbolen (den Index) und den zugehörigen Wortvektor zurückliefert. Hinweis: Implementieren Sie die Prozedur zunächst nur für Anfragen, also Symbollisten, nicht für Dokumente. Verwenden Sie Rekursion. Bei einem leeren Index ist der Wortvektor vollständig und es kann die leere Liste zurückgegeben werden. Bei nicht-leerem Index müssen Sie überprüfen ob das erste Wort im Index in der Symbolliste vorkommt und dementsprechend eine 1 oder 0 vor dem restlichen Wortvektor einfügen. Hierzu können Sie die Prozedur `cons` verwenden. Anschließend fügen Sie zu Beginn der Prozedur eine Behandlung heterogener Daten ein, die im Falle eines Dokumentes als erstem Parameter die Prozedur rekursiv mit dem Dokumenteninhalt aufruft.)
  - Berechnen Sie `(word-vector '(java mouse algol) myindex)`
  - Berechnen Sie `(word-vector doc1 myindex)`

**Lösungsvorschlag:**

s. u. Zeile 84 - 130

Nun benötigen wir noch eine Struktur zur Ausgabe der Suchergebnisse. Diese soll einen Dokumentnamen zusammen mit einem Punktwert, dem `score` enthalten:

```
(define-struct result (name score))
```

5. Implementieren Sie eine Prozedur `compare`, die eine Suchanfrage (eine Liste von Symbolen), ein Dokument und einen Index erhält und daraus ein `result` erzeugt. Wenden Sie zur Berechnung des `score` die Funktion `cosine-similarity` auf den Wortvektor der Suchanfrage und den Wortvektor des Dokumenteninhaltes an. Für die Erstellung des Wortvektors verwenden Sie den übergebenen Index.

**Lösungsvorschlag:**

s. u. Zeile 130- 147



6. Schreiben Sie eine Prozedur `query`, die eine Suchanfrage und eine Liste von Dokumenten übergeben bekommt und eine Liste von Suchergebnissen zurückliefert. Hinweis: Verwenden Sie Rekursion. Sind keine Dokumente in der Liste, so ist das Ergebnis die leere Liste. Ansonsten verwenden Sie die Prozedur `compare`, um das Ergebnis des ersten Dokumentes zu berechnen und `cons` um die restlichen Ergebnisse anzuhängen.

### Lösungsvorschlag:

```

1  ;; Contract: member?: los symbol -> boolean
2  ;; Purpose: computes if symbol is member of los
3  ;; Example: (member? '(a b) 'b) -> true
4  (define (member? los symbol)
5    (cond
6      ;; list empty? anchor reached,
7      ;; do not recurse further
8      ;; symbol not contained in empty list, return false
9      [(empty? los) false]
10     ;; else, list contains at least one element
11     ;; return true if the first element is equal
12     ;; to searched symbol
13     [(symbol=? (first los) symbol) true]
14     ;; first element is not the searched symbol,
15     ;; check if searched symbol is member of
16     ;; the rest of list by the recursive call
17     [else (member? (rest los) symbol)]))
18  ;; Test
19  (check-expect (member? '(a b c) 'c) true)
20
21  ;; Contract: remove-duplicates los -> los
22  ;; Purpose: removes all duplicates from a list of symbols
23  ;; Example: (remove-duplicates '(a a b)) -> '(a b)
24  (define (remove-duplicates list-of-symbols)
25    (cond
26      ;; list empty? anchor reached,
27      ;; do not recurse further
28      ;; empty list does not contain duplicates
29      ;; and can be returned
30      [(empty? list-of-symbols) list-of-symbols]
31      ;; first element is a duplicate
32      ;; (is a member of the rest of the list)
33      [(member? (rest list-of-symbols) (first list-of-symbols))
34       ;; so return rest of the list with remaining
35       ;; duplicates removed
36       (remove-duplicates (rest list-of-symbols))]
37      ;; first element is not a duplicate.
38      ;; Remove duplicates from the rest of the list
39      ;; and add first element at the beginning
40      [else (cons
41              (first list-of-symbols)
42              (remove-duplicates (rest list-of-symbols)))]))
43  ;; Test
44  (check-expect (remove-duplicates '(a a b)) '(a b))
45
46  ;-----
47  ; Aufgabe / Task 5.1
48  ;-----

```

```

49 ;; Contract: vec-mult: list-of-num list-of-num -> num
50 ;; Purpose: computes the inner product of two vectors
51 ;; Example: (vec-mult '(1 0 1) '(1 0 1)) -> 2
52 (define (vec-mult vector1 vector2)
53   (cond
54     ;; list-of-num empty? anchor reached, do not recurse further
55     ;; inner product of two empty vectors is 0
56     [(empty? vector1) 0]
57     ;; else add product of first components to inner product of remaining
58     ;; components
59     [else (+ (* (first vector1) (first vector2)) (vec-mult (rest vector1)
60                                                             (rest vector2)))]))
61
62 ;; Test
63 (check-expect (vec-mult '(1 0) '(0 1)) 0)
64
65 ;; Contract: cosine-similarity list-of-num list-of-num -> num
66 ;; Purpose: computes the cosine of the two vectors
67 ;; Example: (cosine-similarity '(1 0) '(0 1)) -> 0
68 (define (cosine-similarity vector1 vector2)
69   (/
70     (vec-mult vector1 vector2)
71     (*
72       (sqrt (vec-mult vector1 vector1))
73       (sqrt (vec-mult vector2 vector2)))))
74
75 ;; Test
76 (check-expect (cosine-similarity '(1 0) '(0 1)) 0)
77 (check-within (cosine-similarity '(1 1) '(1 1)) 1 0.001)
78
79 ;-----
80 ; Aufgabe / Task 5.2
81 ;-----
82
83 ;; struct for storing documents
84 (define-struct document (name content))
85
86 ;; first document
87 (define doc1
88   (make-document 'doc1 '(mouse keyboard screen)))
89
90 ;; second document
91 (define doc2
92   (make-document 'doc2 '(pascal java scheme)))
93
94 ;-----
95 ; Aufgabe / Task 5.3
96 ;-----
97
98 ;; Contract: index: list-of-documents -> list-of-symbols
99 ;; Purpose: builds a list of all symbols contained in the content
100 ;;           of the documents contained in list-of-documents without
101 ;;           duplicates
102 ;; Example: (list-of-documents
103 ;;           (list
104 ;;             (make-document 'doc1 '(a b))
105 ;;             (make-document 'doc2 '(b c)))) -> '(a b c)
106 (define (index list-of-documents)
107   (cond
108     ;; list-of-documents empty? anchor reached, do not recurse further
109     ;; empty list does not contain any documents and therefore no
110     ;; words return empty list as index

```

```

105     [(empty? list-of-documents) empty]
106     ;; else, append content of first document from list
107     ;; to the index of the rest of documents AND REMOVE DUPLICATES
108     [else (remove-duplicates
109           (append
110             (document-content (first list-of-documents))
111             (index (rest list-of-documents))))])]
112 ;; Test:
113 (check-expect
114   (index (list (make-document 'doc1 '(a b)) (make-document 'doc2 '(b c))))
115   '(a b c))
116
117 ;; index of the two documents doc1 and doc2
118 (define myindex
119   (index (list doc1 doc2)))
120
121
122 ;-----
123 ;   Aufgabe / Task 5.4
124 ;-----
125 ;; Contract: word-vector: document-or-los los -> los
126 ;; Purpose: compute a word-vector for document-or-los
127 ;;           using los as index. If the first argument
128 ;;           is a document, the document contents are used
129 ;;           else the list of symbols is used to check
130 ;;           whether a word from the index occurs.
131 ;; Example: (word-vector '(a c e) '(a b c d e)) -> '(1 0 1 0 1)
132 (define (word-vector document-or-query index)
133   (cond
134     ;; index empty? anchor reached,
135     ;; do not recurse further and return empty word vector
136     [(empty? index) empty]
137     ;; first argument is a document,
138     ;; call recursively with document contents
139     [(document? document-or-query)
140      (word-vector (document-content document-or-query) index)]
141     ;; now first argument is not a document and thus a list of symbols
142     ;; first word from the index is contained -> add 1
143     ;; to the beginning of word vector for remaining index
144     [(member? document-or-query (first index))
145      (cons
146        1
147        (word-vector document-or-query (rest index))))]
148     ;; first word from the index is not contained -> add 0
149     ;; to the beginning of word vector for remaining index
150     [else (cons
151            0
152            (word-vector document-or-query (rest index))))]))
153
154 ;; Test
155 (check-expect
156   (word-vector '(a c e) '(a b c d e))
157   (list 1 0 1 0 1))
158
159 (check-expect
160   (word-vector '(java mouse algol) myindex)
161   (list 1 0 0 0 1 0))
162

```

```

163 (check-expect
164   (word-vector doc1 myindex)
165   (list 1 1 1 0 0 0))
166
167
168
169 ;-----
170 ;   Aufgabe / Task 5.5
171 ;-----
172 ;; struct to store results, document names with a similarity score
173 (define-struct result (name score))
174
175 ;; Contract: compare: los document los -> result
176 ;; Purpose:  computes the similarity for document and query (los)
177 ;;           and returns a result
178 ;; Example:  (query '(java mouse scheme) doc1)
179 ;;           -> (make-result 'doc1 0.111..)
180 (define (compare q document index)
181   (make-result
182     (document-name document)
183     (cosine-similarity
184       (word-vector q index)
185       (word-vector (document-content document) index))))
186
187 ;; Test 1
188 (check-expect
189   (result? (compare (document-content doc1) doc1 myindex))
190   true)
191 (check-expect
192   (result-name (compare (document-content doc1) doc1 myindex))
193   'doc1)
194 (check-within
195   (result-score (compare (document-content doc1) doc1 myindex))
196   1
197   0.01)
198
199 ;; Test 2
200 (check-expect
201   (result? (compare '(java mouse) doc1 myindex))
202   true)
203 (check-expect
204   (result-name (compare '(java mouse) doc1 myindex))
205   'doc1)
206 (check-within
207   (result-score (compare '(java keyboard mouse) doc1 myindex))
208   0.666
209   0.001)
210
211
212
213 ;-----
214 ;   Aufgabe / Task 5.6
215 ;-----
216 ;; Contract: query-help los list-of-documents los -> list-of-results
217 ;; Purpose:  computes the similarity for each document in the
218 ;;           list-of-documents with the list-of-symbols using los as
219 ;;           index
220 ;;           returns them as a list of results

```

```

220 ;; Example: (query-help '(java mouse scheme) (list doc1 doc2) (index
      doc1 doc2)) ->
221 ;;          (list
222 ;;            (make-result 'doc1 0.111..)
223 ;;            (make-result 'doc2 0.333...))
224 (define (query-help q list-of-documents idx)
225   (if (empty? list-of-documents)
226       empty
227       (cons (compare q (first list-of-documents) idx) (query-help q (rest
      list-of-documents) idx))))
228
229 ;; Test 1
230 (define query-help-test1 (query-help '(java mouse scheme) (list doc1 doc2
      ) (index (list doc1 doc2))))
231 (check-expect (cons? query-help-test1) true)
232 (check-expect (result? (first query-help-test1)) true)
233 (check-expect (result? (first (rest query-help-test1))) true)
234 (check-expect (result-name (first query-help-test1)) 'doc1)
235 (check-expect (result-name (first (rest query-help-test1))) 'doc2)
236 (check-within (result-score (first query-help-test1)) 0.333 0.001)
237 (check-within (result-score (first (rest query-help-test1))) 0.666 0.001)
238
239 ;; Test 2
240 (define query-help-test2 (query-help '(keyboard mouse scheme) (list doc1
      doc2) (index (list doc1 doc2))))
241 (check-expect (cons? query-help-test2) true)
242 (check-expect (result? (first query-help-test2)) true)
243 (check-expect (result? (first (rest query-help-test2))) true)
244 (check-expect (result-name (first query-help-test2)) 'doc1)
245 (check-expect (result-name (first (rest query-help-test2))) 'doc2)
246 (check-within (result-score (first query-help-test2)) 0.666 0.001)
247 (check-within (result-score (first (rest query-help-test2))) 0.333 0.001)
248
249
250 ;; Contract: query: los list-of-documents -> list-of-results
251 ;; Purpose: computes the similarity for each document in the
252 ;;           list-of-documents with the list-of-symbols and
253 ;;           returns them as a list of results
254 ;; Example: (query '(java mouse scheme) (list doc1 doc2)) ->
255 ;;          (list
256 ;;            (make-result 'doc1 0.111..)
257 ;;            (make-result 'doc2 0.333...))
258 (define (query q list-of-documents)
259   (query-help q list-of-documents (index list-of-documents)))
260
261 ;; Test
262 (define query-test1 (query '(java mouse scheme) (list doc1 doc2)))
263 (check-expect (cons? query-test1) true)
264 (check-expect (result? (first query-test1)) true)
265 (check-expect (result? (first (rest query-test1))) true)
266 (check-expect (result-name (first query-test1)) 'doc1)
267 (check-expect (result-name (first (rest query-test1))) 'doc2)
268 (check-within (result-score (first query-test1)) 0.333 0.001)
269 (check-within (result-score (first (rest query-test1))) 0.666 0.001)
270
271 ;; Test
272 (define query-test2 (query '(java pascal scheme) (list doc1 doc2)))
273 (check-expect (cons? query-test2) true)

```

```
274 (check-expect (result? (first query-test2)) true)
275 (check-expect (result? (first (rest query-test2))) true)
276 (check-expect (result-name (first query-test2)) 'doc1)
277 (check-expect (result-name (first (rest query-test2))) 'doc2)
278 (check-within (result-score (first query-test2)) 0.0 0.001)
279 (check-within (result-score (first (rest query-test2))) 1 0.001)
```